

## REGION FILL

### PURPOSE

Specifies whether or not regions on subsequent plots are filled. Additionally, it is used to specify whether or not certain types of diagrammatic graphics are filled.

### DESCRIPTION

On a plot, a region is defined as the figure formed by the line connecting points belonging to a common trace and a region base (typically zero). The REGION FILL command can be used to generate a solid fill or a cross-hatch fill of this region. The attributes of the fill are set with additional REGION commands (see the RELATED COMMANDS section below). The attributes of the region border are set with LINE, LINE COLOR, and LINE THICKNESS commands.

The diagrammatic graphics commands CIRCLE, CUBE, DIAMOND, ELLIPSE, HEXAGON, PYRAMID, SEMI-CIRCLE, and TRIANGLE can be filled using the REGION FILL command. The BOX command has its own attribute setting commands.

### SYNTAX

REGION FILL <ON/OFF> <ON/OFF> <ON/OFF> etc.

where ON specifies that the region is filled while OFF specifies that it is not. Up to 100 region fill switches can be specified.

### EXAMPLES

```
REGION FILL ON OFF ON OFF
REGION FILL OFF OFF ON
REGION FILL ON ALL
REGION FILL ALL ON
REGION FILL
```

### NOTE 1

There are some special settings of the REGION FILL switch for the CUBE command. These specify which faces of the cube are to be filled. These options do not apply to any of the other currently supported diagrammatic graphics or plot regions. See the documentation for the CUBE command in the Diagrammatic Graphics chapter for details.

### NOTE 2

The diagrammatic graphics commands use the first setting of the REGION FILL command only.

### NOTE 3

Regions can be used for the following applications in DATAPLOT:

1. They can be used to draw filled or cross-hatched 2d polygons. These polygons are not limited to convex polygons.
2. Statistical maps can be generated. The boundary for each area (e.g., state or county) should be defined as a common trace. This area can then be treated as a polygon. At this time, DATAPLOT does not provide map files. Hopefully, some will be added in future implementations.
3. Area plots can be generated or the area between curves can be filled. Although this type of plot is possible with DATAPLOT, it is generally not a recommended graphical technique.

### NOTE 4

At this time, the region fill capability should not be used for 3d plots. The results will be unpredictable.

### NOTE 5

The REGION FILL command with no arguments sets the region fill to OFF for all traces. The REGION FILL command with the word ALL before or after the specified fill switch assigns that region type to all traces; thus REGION FILL ON ALL or REGION FILL ALL ON plots regions for all traces.

### NOTE 6

The BAR, REGION, SPIKE, CHARACTER, and LINE switch all work independently of each other. That is, a plot point can be a line, a character, a bar, a spike or a region or any combination of the above.

### DEFAULT

All regions off

**SYNONYMS**

None

**RELATED COMMANDS**

PLOT	=	Generates a data or function plot.
REGION BASE	=	Sets the base locations for plot regions.
REGION FILL COLOR	=	Sets the color for region solid fills.
REGION PATTERN	=	Sets the types for region fill patterns.
REGION PATTERN COLOR	=	Sets the color for region hatched fills.
REGION PATTERN LINE	=	Sets the line types for region fill patterns.
REGION PATTERN SPACING	=	Sets the line spacing for region fill patterns.
REGION PATTERN THICKNESS	=	Sets the line thickness for region fill patterns.
LINE COLOR	=	Sets the color for region border lines.
LINE	=	Sets the types for region border lines.
LINE THICKNESS	=	Sets the line thickness for region border lines.

**APPLICATIONS**

Statistical maps, area charts, filled 2d polygons


**IMPLEMENTATION DATE**

Pre-1987

PROGRAM 1

```

MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
TITLE SIZE 3
.
REGION FILL ON OFF
REGION BASE 0 ALL
TITLE NORPDF FROM -1.5 TO -1
PLOT NORPDF(X) FOR X = -1.5 0.01 -1 AND
PLOT NORPDF(X) FOR X = -3 0.01 3
.
TITLE PLOT OF SINE CURVE
PLOT SIN(X) FOR X = -6.28 0.1 6.28
.
LET X = DATA 81 82 83 84 85
LET Y = DATA 2 5 9 15 28
TITLE AREA CURVE
PLOT Y X
.
TITLE PIE CHART
LINE THICKNESS 0.3 ALL
REGION FILL ON ALL
REGION BASE
REGION FILL COLOR G10 G30 G50 G70 G90
PIE CHART Y X
END OF MULTIPLY
    
```



PROGRAM 2

```

MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100; TITLE SIZE 3
SERIAL READ X2
.6255459213E+002 .6408000851E+002 .7055372143E+002 .7249774456E+002 .7415955067E+002
.7837881279E+002 .8329592896E+002 .8584967327E+002 .9298062229E+002 .9053282833E+002
.8058299351E+002 .7574584389E+002 .7132878590E+002 .6534696388E+002 .6277094793E+002
END OF DATA
SERIAL READ Y2
.4360801888E+002 .4019557142E+002 .3765282536E+002 .3541294813E+002 .3788125515E+002
.3731980515E+002 .3269992304E+002 .2901664400E+002 .3339557743E+002 .3760073233E+002
.4461946392E+002 .4518656731E+002 .4718765020E+002 .4567384863E+002 .4543599606E+002
END OF DATA
REGION FILL ON; REGION BASE POLYGON; TITLE PLOT A POLYGON
PLOT Y2 X2
.
SKIP 25
READ TEXAS.DAT X Y
REGION FILL ON; REGION BASE POLYGON
FRAME OFF
TITLE MAP OF TEXAS
PLOT Y X
REGION BASE 0
.
TITLE; PLOT
MOVE 50 97; JUSTIFICATION CENTER
TEXT FILLED DIAGRAMMATIC GRAPHICS REGION FILL ON
CIRCLE 10 10 20 20; REGION FILL ONTS; CUBE 40 70 48 78
REGION FILL ON; ELLIPSE 10 90 15 80 20 90
DIAMOND 80 70 84 65 88 70; HEXAGON 60 10 80 30
END OF MULTIPLY
    
```

